

Often the best way to learn is by example. The sample programs shown here aren't necessarily intended to do anything useful by themselves, but are meant to show how one or more elements of the XPRESS language are used.

Appendix A: Example Programs

Example 1

```
!  
! Blink a direct output at a 2-second rate.  
!  
CONFIG SC = HCS180  
  
DEFINE BlinkTimer = Timer(0)  
DEFINE Light = Output(0)  
  
BEGIN  
IF Reset THEN  
    BlinkTimer = ON  
END  
  
IF BlinkTimer>=2 THEN  
    IFA Light=OFF THEN  
        Light = ON  
    ELSE  
        Light = OFF  
    END  
    BlinkTimer = ON  
END
```

Example 2

```
!  
! Watch a motion detector connected to a netbit and count  
! how many times motion is detected in a 30-second period.  
! If the count exceeds 5 within that time period, sound  
! the horn also connected to a netbit and turn on the  
! lights.  
!  
CONFIG SC = HCS180  
CONFIG PL-Link = 1  
CONFIG DIO = 1  
CONFIG LCD = 1  
  
DISPLAY Modules = L  
  
DEFINE Motion      = Netbit(0)  
DEFINE MotionCount = Variable(0)  
DEFINE MotionTimer = Timer(0)  
DEFINE Light1      = Module(L5)  
DEFINE Light2      = Module(L7)  
DEFINE Horn        = Netbit(7)  
DEFINE Console     = LCD(0)
```

```

BEGIN
IF Reset THEN
    MotionTimer = OFF; Motion = ON
    Horn = OFF; MotionCount = 0
END

IF MotionTimer>30 AND MotionCount>5 THEN
    Horn = ON
    Light1 = ON; Light2 = ON
    MotionTimer = OFF; MotionCount = 0
    Console = "Motion limit exceeded.\n"
END

IFA Motion=ON THEN
    IF MotionTimer=OFF THEN
        MotionTimer = ON
    END
    IF MotionTimer = ON THEN
        Inc(MotionCounter)
    END
END
END

```

Example 3

```

!
! Turn on lights in response to motion, but only if it is
! dark outside. Turn the lights off 20 minutes after the
! last motion has been detected at all hours (so the
! lights are automatically turned off even during the
! day). A light-level detector is connected to an
! ADIO-Link A/D converter.
!
CONFIG SC = HCS180
CONFIG PL-Link = 1
CONFIG ADIO-Link = 1

DISPLAY Modules = E

DEFINE LivingLamp    = Module(E2)
DEFINE LivingFloods = Module(E12)
DEFINE LivingTimer   = Timer(45)
DEFINE LivingMotion  = Input(4)
DEFINE LightLevel    = ADC(16)
DEFINE Setpoint      = 85
DEFINE Mode          = Variable(3)
DEFINE Day           = 0
DEFINE Night         = 1

BEGIN
IF Reset THEN
    Mode = Day; LivingTimer = OFF
END

```

```

IF LightLevel<=Setpoint THEN
    Mode = Night
END
IF LightLevel>Setpoint THEN
    Mode = Day
END
IF Mode=Night AND LivingMotion=EDGE THEN
    LivingLamp = ON; LivingFloods = ON
END

IF LivingMotion=EDGE THEN
    LivingTimer = ON
END

IF LivingTimer>=20 THEN
    LivingLamp = OFF; LivingFloods = OFF
    LivingTimer = OFF
END

```

Example 4

```

!
! A number of buttons on your hand-held remote are defined
! to put the HCS/SS into different modes. Actions are
! carried out depending on those modes.
!
CONFIG SC = HCS180
CONFIG PL-Link = 1
CONFIG IR-Link = 3

DISPLAY Modules = B,D,F

DEFINE MovieMode    = IRcode(2)
DEFINE DinnerMode   = IRcode(3)
DEFINE RomanticMode = IRcode(7)
DEFINE SleepMode    = IRcode(9)
DEFINE FamilyRoom   = 0          ! IR-Link #0
DEFINE DiningRoom   = 1          ! IR-Link #1
DEFINE Bedroom      = 2          ! IR-Link #2
DEFINE FamilyLight   = Module(F3)
DEFINE DiningLight   = Module(D1)
DEFINE BedLight      = Module(B4)
DEFINE FamilyShades  = Output(2)

BEGIN
IF MovieMode=FamilyRoom THEN
    FamilyLight = Dim(8)
    FamilyShades = ON
END

IF DinnerMode=DiningRoom THEN
    DiningLight = ON
END

```

```

IF RomanticMode=DiningRoom THEN
    DiningLight = Dim(10)
END

IF SleepMode=Bedroom THEN
    BedLight = OFF; DiningLight = OFF
    FamilyLight = OFF; FamilyShades = ON
END

```

Example 5

```

!
! The following is a simple telephone-based menuing system
! using the text-to-speech synthesizer for voice prompts
! and the DTMF board for user input.
!
CONFIG SC = HCS180
CONFIG PL-Link = 1

DISPLAY Modules = A,B

DEFINE LightLevel    = ADC(0)
DEFINE RingTimer     = Timer(3)
DEFINE AlarmOn       = Variable(17)
DEFINE Password      = Variable(30)
DEFINE PhoneLevel    = Variable(31)
DEFINE Selection     = Variable(32)
DEFINE Light         = Variable(33)
DEFINE DoorSensor    = Input(4)
DEFINE InsideLight   = Module(A6)
!=====
BEGIN                      ! Start of continuous section

IF Reset THEN
    ClearTimers
    OnHook
    PhoneLevel = 0
    Say"Successful system reset."
END

IF DoorSensor=EDGE AND DoorSensor=ON THEN
    InsideLight = ON
END

IF DoorSensor=OFF THEN
    InsideLight = OFF
END

!=====
SEQUENTIAL                ! Start of sequential section
!=====
!
! Wait for a single ring followed by a hang up.
! Then wait up to one minute for another ring.
! If detected, immediately pick up the phone.

```

```

! This is done so an answering machine can be
! used on the same line as the computer.
!
IF RingTimer=OFF AND Rings=1 THEN
    RingTimer = ON
END

IF RingTimer=ON AND Rings>1 THEN
    RingTimer=OFF
END
IF RingTimer>60 THEN
    RingTimer = OFF
END

IF RingTimer>8 AND Rings=1 THEN
    RingTimer = OFF
    OffHook
    Say"~@ ~60P ~3F"
    Say"~2 Hello. ~5S"
    Say"Welcome to our house."
    Say"Please enter your password."
    Password = DTMFnumber(150)
    IFA Password=1234 THEN
        Say"|Thank you."
        PhoneLevel = 1
    ELSE
        SayW"That is wrong. Goodbye."
        PhoneLevel = 0
        OnHook
    END
END

IF PhoneLevel=1 THEN
    Say"Please make a selection."
    Say"Press zero for a list of choices."
    Selection = DTMFdigit(250)
    IFA Selection<0 THEN
        SayW"Timed out. Good bye."
        PhoneLevel = 0
        OnHook
    END
    IFA Selection=0 THEN
        Say"|Press 1 for light level.."
        Say"Press 2 for alarm status.."
        Say"Press 9 to finish.."
    END
    IFA Selection=9 THEN
        SayW"|Have a nice day. Good bye."
        PhoneLevel = 0
        OnHook
    END
    IFA Selection=1 THEN
        Light = (LightLevel * 100) / 26
        Say"|~D DH IY ~T outside light"
        Say"is at %P1%%. ",Light
    END
END

```

```

    IFA Selection=2 THEN
        IFA AlarmOn=TRUE THEN
            Say"|~D DH IY ~T alarm is on."
        ELSE
            Say"|~D DH IY ~T alarm is off."
        END
    END
    IFA (Selection>2 AND Selection<9)
        OR Selection>9 THEN
            Say"|Invalid selection. Please try again."
        END
    END
END

IF PhoneLevel=0 THEN
    OnHook
END

```

Example 6

```

!
! Allow both a modem and the voice/DTMF pair to occupy the
! same phone line. If you call and allow the phone to ring
! twice, the voice is used to answer the phone and presents
! you with a list of options. If you call, let the phone
! ring once, hang up, then call again within 1 minute, the
! modem will answer instead, allowing the remote use of
! HOST.
!
CONFIG SC = HCS180

DEFINE RingTimer = Timer(3)

BEGIN

IF Reset THEN
    ClearTimers
    ModemRings = 0
END

IF RingTimer=OFF AND Rings=1 THEN
    RingTimer = ON
END

IF RingTimer=ON AND Rings>1 THEN
    RingTimer = OFF
    ModemRings = 0                                ! Disable modem answer
END

IF RingTimer>60 THEN
    RingTimer = OFF
    ModemRings = 0                                ! Disable modem answer
END

```

```
IF RingTimer>8 THEN
    ModemRings = 1                ! Enable modem answer
END
```

SEQUENTIAL

```
IF Rings>=2 THEN
    RingTimer = OFF
    OffHook
    SayW"This is the menu"
    .
    .
    .
    OnHook
END
```

